

**Блоки, изменяющие
маршруты транзактов**

Transfer

1. Режим безусловной передачи

TRANSFER ,B

B – метка блока, на который происходит перенаправление

2. Режим *статистической передачи*

TRANSFER A,B,C

A - вероятность, с которой транзакт направляется в блок, указанный в поле C. С вероятностью $1-A$ транзакт направляется в блок, указанный в поле B (или в следующий блок, если поле B пусто).

Transfer

3. Режим *логической передачи*:

TRANSFER A,B,C

A - BOTH.

Транзакт, поступающий в блок TRANSFER, сначала пытается войти в блок, указанный в поле В (или в следующий блок, если поле В пусто), а если это не удаётся, то в блок, указанный в поле С. Если и эта попытка неудачна, то транзакт задерживается в блоке TRANSFER до изменения условий в модели, делающего возможным вход в один из блоков В или С.

Пример

- На вход системы поступает пуассоновский поток заявок с интенсивностью $\lambda=0.05$ (1/с). Система состоит из двух каналов передачи данных. Если в момент прихода заявки хотя бы один канал свободен, то заявки передаются по этому каналу. Если оба канала заняты, то 90% заявок ждут 10 с и предпринимают повторную попытку передачи, а остальные 10% уничтожаются. Время обслуживания в каждом из каналов распределено по экспоненциальному закону с интенсивностью $\mu=0.025$ (1/с).
- Смоделировать процесс обслуживания 1000 заявок.

Решение

```
SMO2    STORAGE 2
        GENERATE (EXPONENTIAL(1,0,20))
TRY     TRANSFER BOTH,,MET
        ENTER  SMO2
        ADVANCE (EXPONENTIAL(1,0,40))
        LEAVE  SMO2
        TERMINATE 1
MET     TRANSFER .9,,OUT
        ADVANCE 10
        TRANSFER ,TRY
OUT     TERMINATE 1
```

TEST

TEST X A,B,C

X - условие проверки между СЧА:

L (меньше);

LE (меньше или равно);

E (равно);

NE (не равно);

GE (больше или равно);

G(больше).

А и В - сравниваемые СЧА. Если условие AXB выполняется, то TEST пропускает транзакт в следующий блок. Если же это условие не выполняется, то транзакт переходит к блоку, указанному в поле С, а если это поле пусто, то задерживается перед блоком TEST до тех пор, пока логическое выражение AXB не станет истинным.

пример

- Имеется пятиканальная система, на вход которой поступает пуассоновский поток заявок с интенсивностью $\lambda=0.05$ (1/с). Время обслуживания распределено по экспоненциальному закону со средним значением 100 с. В случае, если в момент прихода заявки все каналы заняты, она ожидает в очереди. Длина очереди рассчитана на 3 места. Смоделировать обслуживание 1000 заявок.

Решение

```
Syst      STORAGE 5
          GENERATE (EXPONENTIAL(1,0,20))
          TEST L   Q$LINE,3,OTK
          QUEUE   LINE
          ENTER   Syst
          DEPART  LINE
          ADVANCE (EXPONENTIAL(1,0,100))
          LEAVE   Syst
          TERMINATE 1
OTK       TERMINATE 1
          START   1000
```

GATE

- GATE X A,B
- Вспомогательный операнд X содержит код состояния проверяемого объекта и может принимать следующие значения:
- а) одноканальные устройства U (устройство занято), NU (устройство свободно), I (устройство захвачено), NI (устройство не захвачено);
- б) многоканальные устройства (МКУ) SE (МКУ пусто), SNE (МКУ не пусто), SF (МКУ заполнено), SNF (МКУ не заполнено);
- в) логические переключатели (ЛП) LS (ЛП включен); LR (ЛП выключен).

GATE

- В поле А указывается имя или номер проверяемого объекта. Если проверяемый объект находится в заданном состоянии, то GATE пропускает транзакт к следующему блоку. Если же заданное условие не выполняется, то транзакт переходит к блоку, указанному в поле В, а если это поле пусто, то задерживается перед блоком GATE до тех пор, пока объект А не окажется в состоянии Х.

Пример.

Рассмотрим функционирование простейшей одноканальной системы, на вход которой поступает пуассоновский поток заявок с интенсивностью $\lambda=0.1$ (1/с). Если система свободна, заявка принимается к обслуживанию и обслуживается в среднем 10с (время распределено экспоненциально). В противном случае она получает отказ.

Решение

```
GENERATE (EXPONENTIAL(1,0,10))  
GATE NU    USTR,OTK  
SEIZE     USTR  
ADVANCE(EXPONENTIAL(1,0,10))  
RELEASE   USTR  
TERMINATE 1  
OTK    TERMINATE
```